Universidade do Minho

Departamento de Informática

Modelos Determinísticos de Investigação Operacional

Carlos Preto (a89587)
Maria João Moreira (a89540)
Pedro Veloso (a89557)
Rui Fernandes (a89138)

janeiro 2021

Índice

Introdução

O método do caminho crítico é aplicado a projetos que podem ser decompostos num conjunto de atividades, que se considera terem durações determinísticas, entre as quais existem relações de precedência. As restrições de precedência traduzem o facto de o instante em que se pode dar início a uma dada atividade ter de ser posterior aos instantes em que terminam as atividades que lhe são precedentes.

Neste trabalho irá ser usada uma rede em que as atividades do projeto são representadas por nós. Também é importante referir que é possível, aumentando os recursos aplicados e com custos suplementares, reduzir a duração de uma atividade. A relação entre custos e reduções depende das características da atividade, e pode ser não-linear.

Na variante em análise, cada atividade tem cinco parâmetros adicionais: o primeiro é o valor do custo normal, expresso em unidades monetárias [U.M.], o segundo é o valor de c1, o custo suplementar de reduzir a duração da atividade de uma unidade de tempo [U.T.], expresso em [U.M./U.T.], o terceiro é o valor da máxima redução de tempo a um custo c1, o quarto é o valor de c2, o custo suplementar de reduzir a duração da atividade de uma unidade de tempo [U.T.] após ter aplicado a máxima redução a um custo c1, expresso em [U.M./U.T.], e o quinto é o valor da máxima redução de tempo a um custo c2.

O objetivo do problema é decidir como devem ser reduzidas as durações das atividades, de modo a realizar o projeto na nova duração desejada, com um custo suplementar mínimo.

Parte 0

Definição do novo Grafo e Diagrama de Grantt inicial

Observando os números de inscrição dos membros do grupo, constamos que o maior número de inscrição era 89587. Fazendo pattern matching para o padrão ABCDE e seguindo as regras do enunciado, determinou-se que as atividades 8 e 7 iriam ser removidas da lista de atividades. Uma vez que as atividades 8 e 7 foram removidas, será necessário ajustar a tabela de Precedências. Como é referido no enunciado do projeto, os sucessores de 8 e 7 passarão a ter como novos precedentes os antecessores das atividades 8 e 7.

Tendo em conta os novos dados, será necessário redefinir o nosso Grafo e a nossa Tabela de Precedência.



Atividade	Duração	Precedências
0	4	-
1	6	0
2	7	1,4
3	2	2,5
4	9	0,6
5	4	4,6,10
6	5	-
9	2	6,10,11
10	8	6
11	7	10

Tabela 1: Tabela de Precedências

As restrições do problema, relativas a cada um dos arcos do Grafo, traduzem as relações de precedência entre as atividades. Para uma dada Atividade *j*, o tempo de início dessa mesma atividade de ser posterior ao tempo de conclusão de cada uma das atividades *i* que precedem.

Dado que *ti* designa o tempo de início da atividade *i*, a função *ti* + *di* designa o tempo de conclusão da atividade *i.* O projeto termina no instante de tempo *tf*, quando todas as atividades predecessoras imediatas da atividade fictícia fim estiverem concluídas.

```
/* Objective function */
min: tf;

/* Variable bounds */
arco_i0: t0 >= ti + 0;
arco_i6: t6 >= ti + 0;
arco_01: t1 >= t0 + 4;
arco_04: t4 >= t0 + 4;
arco_12: t2 >= t1 + 6;
arco_23: t3 >= t2 + 7;
arco_3f: tf >= t3 + 2;
arco_42: t2 >= t4 + 9;
arco_45: t5 >= t4 + 9;
arco_53: t3 >= t5 + 4;
arco_5f: tf >= t5 + 4;
arco_64: t4 >= t6 + 5;
arco_65: t5 >= t6 + 5;
arco_69: t9 >= t6 + 5;
arco_610: t10 >= t6 + 5;
arco_9f: tf >= t9 + 2;
arco_10_5: t5 >= t10 + 8;
arco_10_9: t9 >= t10 + 8;
arco_10_11: t11 >= t10 + 8;
arco_11_9: t9 >= t11 + 7;
```

Variables	result
	23
tf	23
t0	0
ti	0
t6	0
t1	4
t4	5
t2	14
t3	21
t5	14
t9	20
t10	5
t11	13

O ficheiro Output indica quais os tempos de início de execução de cada uma das atividades, bem como qual o tempo em todas estão concluídas, que será 23 U.T.. Com estes dados, constrói-se o Diagrama de Grantt inicial.



Parte 1

Como é referido no enunciado, pretende-se que o tempo de execução do projeto seja reduzido em 3 U.T., sendo que o objetivo do Problema será decidir como devem ser reduzidas as durações das atividades, de modo a realizar o projeto na nova duração desejada, que no nosso problema passou a ser 20 U.T., com um custo suplementar mínimo.

É fornecida informação acerca de como completar a tabela que contém os cinco parâmetros adicionais de cada atividade. Uma vez que a atividade 7 foi removida, apenas se tem de completar a informação referente à atividade 9. Esta atividade tem uma duração de 2 U.T. e é referido que esta pode ser realizada com uma duração de 1 U.T., que corresponde a um valor Máximo de redução a Custo 1 de 1 U.T. e um custo adicional(c1) de 200 U.M.. Além disso, também é possível realizar a atividade 9 com uma duração de 0 U.T., que corresponde a um valor Máximo de redução a Custo 2 de 1 U.T., com um custo adicional(c2) de 400 U.M..

Com base nesta informação, obtém-se a seguinte Tabela:

Atividade	Custo Normal	c1	Máx. red. a custo c1	c2	Máx. red. a custo c2
0	400	200	0,5	100	0,5
1	1000	600	1	300	1
2	1400	1000	3	500	1
3	300	200	0,5	100	0,5
4	2000	800	2	400	1
5	1000	1600	0,5	800	0,5
6	800	180	1	90	1
9	300	200	1	400	1
10	1600	1000	0,5	500	0,5
11	1400	600	1	300	1

Tabela 2: Custos e Reduções

Formulação do Problema no LPSolve

Função Objetivo

$$\min \quad \sum_i c1_i r1\_i + c2_i r2\_i$$
$$\forall i \in \{0, 1, 2, 3, 4, 5, 6, 9, 10, 11\}$$

Os valores $c1_i$ e $c2_i$ são, respetivamente, o custo suplementar de reduzir a duração da atividade i de uma unidade de tempo [U.T.] e o custo suplementar de reduzir a duração da atividade i de uma unidade de tempo [U.T.] após ter aplicado a máxima redução a um custo c1. Os valores de *r1_i* e *r2_i* representam o número de reduções que a atividade i teve, associada aos custos C1 e C2, respetivamente.

Restrições

$$\text{suj. a:} \quad t_j \geq t_i - r1\_i - r2\_i + d_i \qquad \begin{array}{l} \forall i \in \{i, 0, 1, 2, 3, 4, 5, 6, 9, 10, 11\} \\ \forall j \in \{0, 1, 2, 3, 4, 5, 6, 9, 10, 11, f\} \end{array}$$

A restrição designa o tempo de conclusão da atividade i após a redução da duração de di para - *r1_i* - *r2_i* + di, onde *tj* representa o tempo de início da atividade j, *ti* o tempo de início da atividade i, *r1_i* e *r2_i* simbolizam a redução dos custos suplementares 1 e 2, respetivamente, e di representa a duração da atividade i.

Seja *decisao_i* a variável associada à possibilidade de existência de uma redução de tempo a um custo C2, sendo esta uma **variável binária**. Sabe-se que apenas se pode aplicar uma redução de tempo a um custo C2 caso se tenha aplicado uma redução máxima de tempo a um custo C1, e que quer C1 como C2 têm limites máximos. Assim, consideremos as variáveis *valorMax_C1_i* e *valorMax_C2_i* como os custos máximos da atividade i, associados às reduções de custo C1 e C2, respetivamente. Assim, é possível adicionar as seguintes restrições de redução:

$$valorMax\_C1\_i \times decisao\_i \leq r1\_i$$

$$r1\_i \leq valorMax\_C1\_i$$

$$r2\_i \leq valorMax\_C2\_i \times decisao\_i$$

$$\forall i \in \{0, 1, 2, 3, 4, 5, 6, 9, 10, 11\}$$

De maneira a verificar a veracidade das restrições, consideremos os 2 valores possíveis da variável decisao_i:

- ✓ ***Não pode haver redução de tempo a um custo C2 (decisao_i = 0)***

  Considere-se que, numa dada atividade, não pode haver uma redução de tempo a um custo C2 (decisao_i = 0). Então ter-se-á, *r2_i* <= *0*, e como se está a trabalhar com números não negativos*, r2_i = 0*, que simboliza a não aplicação da redução de tempo a um custo C2. Comprova-se que *0* <= *r1_i* <= *valorMax_C1_i*, ou seja, pode haver ou não redução de tempo a um custo C1, o que respeita os requisitos.

- ✓ ***Pode haver redução de tempo a um custo C2 (decisao_i = 1)***

  Considere-se que, numa dada atividade, pode haver uma redução de tempo a um custo C2 (decisao_i = 1). Então ter-se-á, pelas restrições apresentadas acima, que *r2_i* <= *valorMax_C2_i*, simbolizando que pode haver redução de tempo a custo C2. Também se pode comprovar que *valorMax_C1_i* <= *r1_i* <= *valorMax_C1_i,* ou seja, r1_i = valorMax_C1_i, que simboliza que houve a redução máxima de tempo a um custo C1. Com estes valores, fica provado que para haver uma redução de tempo a um custo C2, implica que anteriormente tenha havido uma redução máxima de tempo a um custo C1**,** tal como pretendido no enunciado.

Além destas restrições, é necessário indicar que o ***tf <= 20***, uma vez que é esse o novo tempo máximo permitido para terminar o projeto. Tendo definido a nossa função objetivo e todas as restrições, pode-se formular o ficheiro input do problema, que se encontra na secção de Anexos no fim do relatório.

Solução e verificação do resultado

Após a compilação do ficheiro input acima representado, foi gerado um ficheiro Output, que se encontra na secção de Anexos no fim do relatório . O ficheiro informa acerca do valor das nossas variáveis. Verifica-se que:

- ✓ r1_0 = 0,5
- ✓ r2_0 = 0,5
- ✓ r1_3 = 0,5
- ✓ r2_3 = 0,5
- ✓ r1_6 = 1
- ✓ r2_6 = 1

ou seja, sabe-se que as atividades **0**, **3** e **6** tiveram redução máxima de tempos associadas a ambos os custos, e que passaram a ter um tempo de execução de 3, 1, e 3, respetivamente. Tal resultado pode ser explicado pelas decisões,

- ✓ decisao_0 = decisao_3 = decisao_6 = 1

que demonstram que pode haver redução de tempo a um custo C2 nas atividades 0, 3 e 6. Na solução ótima correspondem a uma redução máxima de tempo a um custo C2.

O ficheiro Output fornece também o custo suplementar mínimo associado à redução dos tempos de execução das atividades 0,3 e 6, sendo este de **570 U.M**. Tal valor pode ser comprovado pelos cálculos:

0,5*200 + 0.5*100 + 0.5*200 + 0.5*100 + 1*180 + 1*90 = 570 U.M.

Também informa quais os novos tempos de início das atividades, de maneira à duração do projeto ser no máximo 20 U.T, permitindo desenhar o Diagrama de Grantt.

- ✓ ti = t0 = t6 = 0
- ✓ t1 = t4 = t10 = 3
- ✓ t11 = 11
- ✓ t2 = 12
- ✓ t5 = 15
- ✓ t9 = 18
- ✓ t3 = 19
- ✓ tf = 20



Conclusão

Através da utilização do Software LPSolve, verificou-se que de maneira a reduzir o tempo de execução total do projeto, será necessário reduzir os tempos de execução das atividades 0, 3 e 6 em respetivamente 1 U.T.(0,5 U.T. associado ao custo suplementar 1 + 0,5 associado ao custo suplementar 2), 1 U.T. (0,5 U.T. associado ao custo suplementar 1 + 0,5 associado ao custo suplementar 2), e 2 U.T. (1 U.T. associado ao custo suplementar 1 + 1 associado ao custo suplementar 2). Com base nessas reduções, ter-se-á um custo suplementar mínimo de 570 U.M.

No geral, o grupo não demonstrou muitas dificuldades na realização do projeto, sendo que a única adversidade encontrada foi na construção do ficheiro input do LPSolve, uma vez que o facto da redução de custo C2 apenas poder ser aplicada após se ter aplicado as reduções máximas de custo C2 levou a que numa fase inicial não se estivesse a considerar as restrições binárias. Após se ter ultrapassado essa adversidade, tornou-se tudo mais simples, permitindo chegar à solução correta do problema proposto.

Anexos

Input LPSolve

```
/* Objective function */
min: 200r1_0 + 100r2_0 +
     600r1_1 + 300r2_1 +
     1000r1_2 + 500r2_2 +
     200r1_3 + 100r2_3 +
     800r1_4 + 400r2_4 +
     1600r1_5 + 800r2_5 +
     180r1_6 + 90r2_6 +
     200r1_9 + 400r2_9 +
     1000r1_10 + 500r2_10 +
     600r1_11 + 300r2_11;

/* Variable bounds */
// tempo máximo para concluir o projecto
tf <= 20;

// relações de precedência
arco_i0: t0 >= ti + 0;
arco_i6: t6 >= ti + 0;
arco_01: t1 >= t0 - r1_0 - r2_0 + 4;
arco_04: t4 >= t0 - r1_0 - r2_0 + 4;
arco_12: t2 >= t1 - r1_1 - r2_1 + 6;
arco_23: t3 >= t2 - r1_2 - r2_2 + 7;
arco_3f: tf >= t3 - r1_3 - r2_3 + 2;
arco_42: t2 >= t4 - r1_4 - r2_4 + 9;
arco_45: t5 >= t4 - r1_4 - r2_4 + 9;
arco_53: t3 >= t5 - r1_5 - r2_5 + 4;
arco_5f: tf >= t5 - r1_5 - r2_5 + 4;
arco_64: t4 >= t6 - r1_6 - r2_6 + 5;
arco_65: t5 >= t6 - r1_6 - r2_6 + 5;
arco_69: t9 >= t6 - r1_6 - r2_6 + 5;
arco_610: t10 >= t6 - r1_6 - r2_6 + 5;
arco_9f: tf >= t9 - r1_9 - r2_9 + 2;
arco_10_5: t5 >= t10 - r1_10 - r2_10 + 8;
arco_10_9: t9 >= t10 - r1_10 - r2_10 + 8;
arco_10_11: t11 >= t10 - r1_10 - r2_10 + 8;
arco_11_9: t9 >= t11 - r1_11 - r2_11 + 7;
```

```
//restrições de redução
0.5 decisao_0 <= r1_0;
r1_0 <= 0.5;
r2_0 <= 0.5 decisao_0;

1 decisao_1 <= r1_1;
r1_1 <= 1;
r2_1 <= 1 decisao_1;

3 decisao_2 <= r1_2;
r1_2 <= 3;
r2_2 <= 1 decisao_2;

0.5 decisao_3 <= r1_3;
r1_3 <= 0.5;
r2_3 <= 0.5 decisao_3;

2 decisao_4 <= r1_4;
r1_4 <= 2;
r2_4 <= 1 decisao_4;

0.5 decisao_5 <= r1_5;
r1_5 <= 0.5;
r2_5 <= 0.5 decisao_5;

1 decisao_6 <= r1_6;
r1_6 <= 1;
r2_6 <= 1 decisao_6;

1 decisao_9 <= r1_9;
r1_9 <= 1;
r2_9 <= 1 decisao_9;

0.5 decisao_10 <= r1_10;
r1_10 <= 0.5;
r2_10 <= 0.5 decisao_10;

1 decisao_11 <= r1_11;
r1_11 <= 1;
r2_11 <= 1 decisao_11;
```

```
bin decisao_0, decisao_1, decisao_2, decisao_3,
    decisao_4, decisao_5, decisao_6, decisao_9,
    decisao_10, decisao_11;
```

Output LPSolve

Variables	MILP Feasible	result
	570,000000000002	570,000000000002
ti	0	0
tf	20	20
t9	18	18
t6	0	0
t5	15	15
t4	2,99999999999999	2,99999999999999
t3	19	19
t2	12	12
t11	11	11
t10	2,99999999999999	2,99999999999999
t1	2,99999999999999	2,99999999999999
t0	0	0
r2_9	0	0
r2_6	1	1
r2_5	0	0
r2_4	0	0
r2_3	0,5	0,5
r2_2	0	0
r2_11	0	0
r2_10	0	0
r2_1	0	0
r2_0	0,5	0,5
r1_9	0	0
r1_6	1	1
r1_5	0	0
r1_4	0	0
r1_3	0,5	0,5
r1_2	0	0
r1_11	0	0
r1_10	0	0
r1_1	0	0
r1_0	0,500000000000005	0,500000000000005
decisao_9	0	0
decisao_6	1	1
decisao_5	0	0
decisao_4	0	0
decisao_3	1	1
decisao_2	0	0
decisao_11	0	0
decisao_10	0	0
decisao_1	0	0
decisao_0	1	1